# Maintainer: Antonio Rojas <arojas@archlinux.org>
# Maintainer: Felix Yan <felixonmars@archlinux.org>
# Contributor: Andrea Scarpino <andrea@archlinux.org>

pkgname=kmousetool
pkgver=18.08.1
pkgrel=2
arch=(x86_64)
pkgdesc='Clicks the mouse for you, reducing the effects of RSI'
url='https://kde.org/applications/utilities/kmousetool/'
license=(GPL FDL)
groups=(kde-applications kdeaccessibility)
depends=(knotifications kxmlgui phonon-qt5 hicolor-icon-theme)
makedepends=(extra-cmake-modules kdoctools libxt)
conflicts=(kdeaccessibility-kmousetool)
replaces=(kdeaccessibility-kmousetool)
source=("https://download.kde.org/stable/applications/$pkgver/src/$pkgname-$pkgver.tar.xz"{,.sig})
sha256sums=('9fdfe3ccf7cb46be83ac795daddcb26a8d9793e3a020502a27b20f9f75fb3737'
            'SKIP')
validpgpkeys=(CA262C6C83DE4D2FB28A332A3A6A4DB839EAA6D7  # Albert Astals Cid <aacid@kde.org>
              F23275E4BF10AFC1DF6914A6DBD2CE893E2D1C87) # Christoph Feck <cfeck@kde.org>

prepare() {
  mkdir -p build
}

build() {
  cd build
  cmake ../$pkgname-$pkgver \
    -DCMAKE_INSTALL_PREFIX=/usr \
    -DBUILD_TESTING=OFF
  make
}

package() {
  cd build
  make DESTDIR="$pkgdir" install
}
